



DECUS

PROGRAM LIBRARY

DECUS NO.	8-71
TITLE	Perpetual Calendar
AUTHOR	E. Singer
COMPANY	McGill University Montreal, Quebec, Canada
DATE	April 18, 1967
SOURCE LANGUAGE	

ATTENTION

This is a USER program. Other than requiring that it conform to submittal and review standards, no quality control has been imposed upon this program by DECUS.

The DECUS Program Library is a clearing house only; it does not generate or test programs. No warranty, express or implied, is made by the contributor, Digital Equipment Computer Users Society or Digital Equipment Corporation as to the accuracy or functioning of the program or related material, and no responsibility is assumed by these parties in connection therewith.

PERPETUAL CALENDAR

Program Library Write-up

DECUS No. 8-71

This program is designed as a computer demonstration. When a valid date is fed into the computer, the corresponding day of the week is typed out. If an invalid date is fed in, "You Goofed, Try Again" is typed out.

The method of computation is based on the Gregorian Calendar and is, therefore, limited to years between 1500 and 4095. The upper limit is due to the capacity of the computer.

The actual computational part of the program is found in locations 20-563. The routine to check the validity of a date is found in locations 564-777. In order to handle years which when converted to octal are considered negative, a small subroutine was placed in page zero. The routine to print out instructions is located in 1000-1333.

This program is loaded with the binary loader. The starting address is 200. When started for the first time, the program will print out instructions on how to feed in a date. After five dates have been processed, the instructions will again be printed out. In order to avoid the instructions being printed out every five times, deposit 0 in location 112.

MINIMUM HARDWARE

4K Storage, ASR-33 Teletype

STORAGE REQUIREMENT

20 - 1333

```

/PERPETUAL CALENDAR
*20
TEMP1,0
TEMP2,0
TEMP3,0
TEMP4,0
MONTH,0
DAY,0
YEAR,0
CENT,0
M260,-260
P12,12
P1750,1750
P144,144
M2,-2
M1,-1
M144,-144
MISK,3777
MASK,7774
PNTR,MON-1
P106,106
M7,-7
P7,7
MON,1;4;3;6;1;4;6;2;5;0;3;5
READP,READ
PRINTP,PRINT
SPAC,240
CR,215
LF,212
DPRINP,DPRINT
BASE,450;461;470;477;507;521;532
BASEP,BASE
GOOFP,600
M1500,-2734
M14,-14
M37,-37
M40,-40
MESK,1
M35,-35
M36,-36
M400,-620
INST,1000
INIT,202
CNTR,-1
M6,-6
LEAD,GOOF+2
LEADER,DIREC-1
RIG,0          /SUBROUTINE IF YEAR > 3777 OCTAL
TAD YEAR
TAD M3720
DCA YEAR
TAD P24
DCA TEMP2
JMP I BIG
M3720,-3720
P24,24
NEWPAG,400
DAYPNT,DAYTST

```

```

*200          /START OF PROGRAM
TLS
JMP I Z INST      /PRINT THE INSTRUCTIONS
START, JMS I Z READP      /READ MONTH
JMS I Z PRINTP
TAD Z M260
DCA Z TEMP1
JMS I Z READP
JMS I Z PRINTP
TAD Z M260
DCA Z TEMP2
TAD Z TEMP1      /CONVERT TO OCTAL
SNA
JMP ZERO
CIA
DCA Z TEMP1
SIX, TAD Z P12
ISZ Z TEMP1
JMP SIX
ZERO, TAD TEMP2
DCA Z MONTH      /DEPOSIT IN MONTH
TAD Z SPAC
JMS I Z PRINTP
JMS I Z PRINTP
CLA
JMS I Z READP      /READ DAY
JMS I Z PRINTP
TAD Z M260
DCA Z TEMP1
JMS I Z READP
JMS I Z PRINTP
TAD Z M260
DCA Z TEMP2
TAD Z TEMP1      /CONVERT TO OCTAL
SNA
JMP FIRST
CIA
DCA Z TEMP1
SEC, TAD Z P12
ISZ Z TEMP1
JMP SEC
FIRST, TAD Z TEMP2
DCA Z DAY      /DEPOSIT IN DAY
TAD Z SPAC
JMS I Z PRINTP
JMS I Z PRINTP
CLA
JMS I Z READP      /READ YEAR
JMS I Z PRINTP
TAD Z M260
DCA Z TEMP1
JMS I Z READP
JMS I Z PRINTP
TAD Z M260
DCA Z TEMP2
JMS I Z READP
JMS I Z PRINTP
TAD Z M260
DCA Z TEMP3

```

```

JMS I Z READP
JMS I Z PRINTP
TAD Z M260
DCA Z TEMP4
TAD Z TEMP1          /CONVERT TO OCTAL
CIA
DCA Z TEMP1
THIRD,TAD Z P1750
ISZ Z TEMP1
JMP THIRD
DCA Z TEMP1
TAD Z TEMP2
CIA
DCA Z TEMP2
FOUR,TAD Z P144
ISZ Z TEMP2
JMP FOUR
DCA Z TEMP2
TAD Z TEMP3
CIA
DCA Z TEMP3
FIVE,TAD Z P12
ISZ Z TEMP3
JMP FIVE
TAD Z TEMP1
TAD Z TEMP2
TAD Z TEMP4
DCA Z YEAR          /DEPOSIT IN YEAR
TAD Z SPAC
JMS I Z PRINTP
JMS I Z PRINTP
CLA
JMP I Z GOOFP       /CHECK DATE FOR VALIDITY
TAD Z MONTH         /IF JAN,FEB
TAD Z M2            /SUBTRACT 1 FROM THE YEAR
SPA SNA
JMP NOTOK
JMP OK
NOTOK,CLA
TAD Z YEAR
TAD Z M1
DCA Z YEAR
OK,CLA
DCA Z TEMP2
TAD Z YEAR
SPA
JMS Z BIG
LOOP1,TAD Z M144
ISZ Z TEMP2
SMA
JMP LOOP1
TAD Z P144          /REMAINDER AFTER DIV. BY 100
DCA Z YEAR
TAD Z TEMP2
TAD Z M1
DCA Z CENT          /CENTURY

```

```

TAD Z YEAR
RTR CLL
AND Z MISK /DIVIDE YEAR BY 4
TAD Z YEAR
DCA Z TEMP1
TAD Z CENT
AND Z MASK
CIA
TAD Z CENT /REMAINDER OF CENTURY AFTER DIV. BY 4
CLL RAL /MULTIPLY BY TWO
CIA
TAD Z TEMP1
DCA Z TEMP1
JMP I NEWPAG /GOES TO NEXT PAGE
*400
TAD Z PNTR
TAD Z MONTH
DCA Z PNTR
TAD I Z PNTR /GET VALUE FOR MONTH
TAD Z TEMP1
TAD Z DAY
TAD Z P106 /MAKE SURE IT IS >0
DCA Z TEMP1
TAD Z MONTH /RESET PNTR
CIA
TAD Z PNTR
DCA Z PNTR
TAD Z TEMP1
LOOP2, TAD Z M7 /GET REMAINDER AFTER DIV. BY 7
SMA
JMP LOOP2
TAD Z P7
DCA Z TEMP1
TAD Z BASEP /GET CORRECT DAY FROM BASE
TAD Z TEMP1
DCA Z BASEP
JMS I Z DPRINP /PRINT DAY
TAD Z CR
JMS I Z PRINTP
CLA
TAD Z LF
JMS I Z PRINTP
CLA
JMP I RETURN /START OVER
RETURN, START-1
READ, 0 /READ ROUTINE
KCC
RWAIT, KSF
JMP RWAIT
KRR
JMP I READ
PRINT, 0 /PRINT ROUTINE
PWAIT, ISF
JMP PWAIT
TLS
JMP I PRINT

```

```

S;A;T;U;R;D;A;Y;0
S;U;N;D;A;Y;0
M;O;N;D;A;Y;0
T;U;E;S;D;A;Y;0
W;E;D;N;E;S;D;A;Y;0
T;H;U;R;S;D;A;Y;0
F;R;I;D;A;Y;0
DPRINT,0          /ROUTINE TO PRINT DAY
TAD I Z BASEP
DCA Z 10
LOOP3,TAD I Z 10
SNA
JMP END
DWAIT,TSF
JMP DWAIT
TLS
CLA
JMP LOOP3
END,TAD Z TEMP1   /RESET BASE
CIA
TAD Z BASEP
DCA Z BASEP
JMP I DPRINT
DAYTST,CLA        /ERROR TEST FOR 0 DAY AND 0 MONTH
TAD Z DAY
SNA CLA
JMP I BAD          /PRINT ERROR MESSAGE
TAD Z MONTH
SNA CLA
JMP I BAD          / " " "
JMP I GOOD         /CONTINUE ERROR ROUTINE
RAD,GOOF
GOOD,SEVEN-5
*600              /ERROR ROUTINE
TAD Z YEAR
SPA
JMP NGOOF          /TEST FOR YEAR >3777 OCTAL
/YES,CONTINUE ROUTINE
TAD Z M1500
SPA SNA
JMP GOOF          /YES, PRINT ERROR MESSAGE
NGOOF,CLA         /NO, TEST IF YEAR <1500 DECIMAL
TAD Z MONTH
TAD Z M14
SMA SZA
JMP GOOF          /NO, PRINT ERROR MESSAGE
JMP I Z DAYPNT    /YES, CHECK FOR ZERO DAY OR MONTH
TAD Z MONTH
TAD Z M7
SPA SNA
JMP SEVEN
JMP EIGHT
SEVEN,CLA
TAD Z MONTH
AND Z MESK
SZA
JMP NINE
JMP TEN
NINE,CLA
TAD Z DAY
TAD Z M40

```

SMA
JMP GOOF
JMP FIN2 /VALID DATE
TEN, TAD Z DAY
TAD Z M37
SMA CLA
JMP GOOF
JMP FIN1 /FEBRUARY TEST
EIGHT, CLA
TAD Z MONTH
AND Z MESK
SZA
JMP ELEV
JMP TWEL
ELEV, CLA
TAD Z DAY
TAD Z M37
SMA
JMP GOOF
JMP FIN2 /VALID DATE
TWEL, TAD Z DAY
TAD Z M40
SMA
JMP GOOF
JMP FIN2 /VALID DATE
FIN1, TAD Z MONTH /FEBRUARY TEST
TAD Z M2
SZA
JMP FIN2 /NO, VALID DATE
TAD Z YEAR /YES, CHECK FOR LEAP YEAR
AND Z MASK
CIA
TAD Z YEAR
SZA
JMP NLEAP /NO
JMP LEAP /YES
NLEAP, CLA
TAD Z DAY
TAD Z M35
SMA
JMP GOOF
JMP FIN2 /VALID DATE
LEAP, TAD Z YEAR /CHECK IF DIVISIBLE BY 100
SPA
TAD Z M3720
TAD Z M144
SMA SZA
JMP LEAP+3
SNA CLA
JMP CLEAP /YES, CHECK IF LEAP CENTURY
DLEAP, TAD Z DAY /NO, LEAP YEAR TEST
TAD Z M36
SMA
JMP GOOF
JMP FIN2 /VALID DATE
CLEAP, CLA /TEST IF DIVISIBLE BY 400
TAD Z YEAR
SPA


```

TAD Z M3720
TAD Z M400
SMA SZA
JMP CLEAP+4
SZA
JMP NLEAP           /NO, NON LEAP YEAR TEST
JMP DLEAP           /YES, LEAP YEAR TEST
FIN2,CLA           /RETURN TO MAIN PROGRAM WITH VALID DATE
JMP I ERR
ERR,333
MESS,COMMNT-1
COMMNT,Y;O;U;SPACE;G;O;O;F;E;D;SPACE;T;R;Y;SPACE;A;G;A;II;N;0
GOOF,CLA           /ERROR MESSAGE AND INSTRUCTION PRINT ROUTINE
TAD MESS
DCA Z 10
LOOP4,TAD I Z 10
SNA
JMP I ARROW
GWAIT,TSF
JMP GWAIT
TLS
CLA
JMP LOOP4
ARROW,426
A=301
B=302
C=303
D=304
E=305
F=306
G=307
H=310
II=311
J=312
K=313
L=314
M=315
N=316
O=317
P=320
Q=321
R=322
S=323
T=324
U=325
V=326
W=327
Y=331
CRR=215
LFF=212
SPACE=240
PERIOD=256
COMMA=254

```

```

*1000
ISZ Z CNTR          /INSTRUCTION ROUTINE
JMP I Z INIT
CLA
TAD Z M6
DCA Z CNTR
TAD Z LEADER
JMP I Z LEAD       /PRINT INSTRUCTIONS
DIRFC,CRR;LFF;LFF;T;H;II;S;SPACE;P;R;O;G;R;A;M;SPACE;II;S;SPACE
D;E;S;II;G;N;E;D;SPACE;T;O;SPACE;G;II;V;E;SPACE
T;H;E;SPACE;D;A;Y;SPACE;O;F;SPACE;T;H;E;SPACE;W;E;E;K
CRR;LFF;W;H;E;N;SPACE;II;T;SPACE;II;S;SPACE;G;II;V;E;N;SPACE
A;SPACE;D;A;T;E;PERIOD;CRR;LFF
T;Y;P;E;SPACE;T;W;O;SPACE;D;II;G;II;T;S;SPACE;F;O;R
SPACE;T;H;E;SPACE;M;O;N;T;H;COMMA;SPACE;T;H;E;N;SPACE;T;W;O
SPACE;D;II;G;II;T;S;SPACE;F;O;R;SPACE;T;H;E;SPACE;D;A;Y;COMMA;CRR;LFF
T;H;E;N;SPACE;F;O;U;R;SPACE;D;II;G;II;T;S;SPACE;F;O;R
SPACE;T;H;E;SPACE;Y;E;A;R;PERIOD;CRR;LFF;T;H;E;SPACE
Y;E;A;R;SPACE;S;H;O;U;L;D;SPACE;B;E;SPACE;L;E;S;S;SPACE
T;H;A;N;SPACE;264;260;260;260;PERIOD;CRR;LFF;0
$

```

